

VOLTAGE COMPARATORS AND SCHMITT TRIGGERS: INTRODUCTION TO BEHAVIORAL MODELING

Version 1 - March 1995

TABLE

Introduction	p1
How does a comparator work?	p1
Starting point	p2
First step: a macro-model description	p4
Second step: what about a non symmetric threshold?	p6
Third step: use of a behavioral model	p9
Last step: a continuous model for hysteresis function	p12
Reference	p15

Introduction

This paper presents an overview of comparator's descriptions and simulation results with SMASHTM. Triggers, which use hysteresis, will be introduced as a generalization of comparators. The purpose of this application note is to show different approaches for modeling such comparators and triggers:

- “electrical” macro-modeling,
- “functional” macro-modeling,
- behavioral modeling, which does not only implement a generic facility (just like macro-model) but also provides a “natural mathematical description” in C code.

How does a comparator work?

We will refer to a general differential comparator, with inputs v_{in} and v_{ref} , and an output v_{out} . The v_{out} signal will switch between v_{outmax} and v_{outmin} , according to the value of the differential input ($v_{in}-v_{ref}$) and threshold value. For a rising or a falling edge, threshold can be different. This alternative path, illustrated on the figure 1, is modeled thanks to an hysteresis function.

An input voltage offset (VIO) can also be taken into account to modify the differential input and then $V=v_{in}-v_{ref}-VIO$ becomes the actual input.




Figure 1: Functional schematic of a comparator

Starting point

The very first example is an electrical model provided by Texas Instrument™. The only tunable parameter is *voutmax*. The following circuit design shows that the output signal is driven by a transistor's collector, connected to a pullup resistor R1. Note that this transistor has a grounded emitter and thus limits *voutmin* at zero. The circuit schematic is displayed below, followed by the corresponding electrical description file.




Figure 2: Comparator with circuit LM211

Electrical description:

```
* LM211 VOLTAGE COMPARATOR "MACROMODEL" SUBCIRCUIT
* CONNECTIONS: NON-INVERTING INPUT
*               | INVERTING INPUT
*               | | POSITIVE POWER SUPPLY
*               | | NEGATIVE POWER SUPPLY
*               | | OPEN COLLECTOR OUTPUT
*               | | OUTPUT GROUND
*
*               | 1 2 3 4 5 6
.SUBCKT LM211 1 2 3 4 5 6
F1 9 3 V1 1
IEE 3 7 DC 100.0E-6
V11 21 1 DC .45
V12 22 2 DC .45
Q1 9 21 7 QIN
Q2 8 22 7 QIN
Q3 9 8 4 QMO
Q4 8 8 4 QMI
.MODEL QIN PNP(IS=800.0E-18 BF=666.7)
.MODEL QMI NPN(IS=800.0E-18 BF=1002)
.MODEL QMO NPN(IS=800.0E-18 BF=1000 CJC=1E-15 TR=102.5E-9)
E1 10 6 9 4 1
V1 10 11 DC 0
Q5 5 11 6 QOC
.MODEL QOC NPN(IS=800.0E-18 BF=103.5E3 CJC=1E-15 TF=11.60E-12 TR=48.19E-9)
DP 4 3 DX
RP 3 4 6.667E3
.MODEL DX D(IS=800.0E-18)
```

Simulation files specify the parameters and results are displayed below.

```
*-----file texas.nsx-----
*COMPARATOR supplied by TEXAS

* Modele LM211:
* Vref gives the threshold input
* voltage.
* Output signal vout, switches
* from 0 to vp.

*comparator
Xtest vin vref vp vm vout 0
+ LM211

*Polarization of the open collector
R1 vout vp 10K
C1 vout 0 1p
```

```
*-----FILE TEXAS.PAT-----

*SOURCES
V_VIN VIN 0 SIN( 0 2 500K 0 0 0 )
V_VREF VREF 0 0.5
V_VP VP 0 5
V_VM VM 0 -5

*LIBRARY
.LIB LM211.CKT


*PARAMETRES FOR LM211.
.EPS      1U 100M 1N
.H        5N 100F 10N 250M 2
.TRAN     15N 4U 0


*SCREEN
.TRACE TRAN V(VIN) V(VOUT) MIN=-
2.6998566E+000 MAX=5.7001629E+000
.TRACE TRAN V(XTEST_11) MIN=-
5.1379340E-002 MAX=7.8030997E-001
.TRACE TRAN IB(Q3_XTEST)
IB(Q5_XTEST) MIN=-1.1959685E-004
MAX=1.1980044E-004
```

In order to compare simulation results, parameters will remain constant in the different following examples. We present in this paper only the particular results for each modeling style, leaving all necessary information in the simulation files to let you run each case.

In the present example, two simulations are run. The first one presents the ideal case of a comparator (vref is grounded and there is no offset). when the second one (refereed as « standard » in the netlist file) enlightens the role of the differential input, namely the influence of vref.

Ideal case



Standard case: vref influence

This kind of electrical model allows accurate simulation of a particular circuit, but it does not take into account general features (such as v_{outmin}). As a consequence, each time we want to test a comparator we must find out the corresponding model, and unfortunately it is not always supplied by manufacturers. It is thus interesting to have a look on a generic model.

First step: a macro-model description

We now introduce a macro-model of a differential comparator from [ref 1]. In the following example, all parameters can be tuned to reach a satisfactory description of any particular comparator. In addition to previous parameters, output can be inverted thanks to the INV parameter.

Note that for a simple comparator, hysteresis is not taken into account.

This macro-model description allows an easier decomposition of the circuit into functional stages than the electrical model does.




Figure 3: Functional stages

The subcircuit is listed below in the netlist file.

```
*-----file comp.nsx-----
*Differential comparator with offset
* VIO          voltage input offset
* GAIN
* VOUTMIN      min output voltage
* VOUTMAX      max output voltage
* RIN          input resistor
* ROUT        output resistor
* INV          = 1 invensor
*              = -1 non-inversor

.SUBCKT COMP vin vref vout \
+ VIO GAIN VOUTMIN VOUTMAX
+ RIN ROUT INV

R1 vin vref RIN
E1 2 0 POLY(3) vin 0 vref 0 4 0
+ 'VIO*GAIN'
+ '-GAIN'
+ GAIN
+ 1e-5
Rtest 2 0 1

E2 3 0 2 0 1MEG
R2 3 4 1T
C1 4 0 1e-15
D1 4 5 SWITCH
V1 5 0 '(VOUTMAX-VOUTMIN)/2'
D2 6 4 SWITCH
V2 6 0 '-(VOUTMAX-VOUTMIN)/2'
.model SWITCH D(IS=10u N=0.1)

EOUT 7 0 POLY(1) 4 0
+ 'VOUTMAX/2+VOUTMIN/2'
+ INV
R3 vout 0 1G
R4 7 vout ROUT
.ENDS COMP

*Standard simulation or ideal if vref=0
Xstandard vin vref vout \
+ 0.0 5.0 -5.0 5.0 1MEG 1K -1
+ COMP

*inversor (only INV is modified)
Xinv vin vref voutinv \
+ 0.0 5.0 -5.0 5.0 1MEG 1K 1
+ COMP

*non-inversor, VIO is a threshold
* As the gain is low, the symetry
* is lost.
Xthreshold vin vref voutvio \
+ 0.5 1.0 -5.0 5.0 1MEG 1K -1
+ COMP

*symetric output thanks to a
* high gain even with a input
* offset.
Xgain vin vref voutgain \
+ 0.5 100.0 -5.0 5.0 1MEG 1K -1
+ COMP
```

```
*-----file comp.pat-----

*SOURCES
V_VIN VIN 0 SIN( 0 2 1MEG 0 0 0 )
V_VREF VREF 0 0.0


.EPS      1U 100M 1N
.H        10N 1F 10N 250M 2
.TRAN     1N 3U 0
.OP EPS_V=1U VMIN=-50MEG VMAX=50MEG
DELTAV=0 EPS_I=100P MAXITER=500

* SCREEN
.TRACE TRAN V(VIN) V(VOUT)
V(VOUTGAIN) V(VOUTVIO) MIN=-
5.9999880E+000 MAX=5.9999285E+000
```

Please note that all simulations are run simultaneously by the simulator (four X-modules are called in the netlist file). This explains a slower convergence analysis compared to the electrical model, but this way, all outputs can be displayed and compared with a single simulation run.


We now present two selected outputs.

Inverted simulation (INV)



The functional macro-model allows us to get an inverted output with a simple modification of the INV parameter (INV= -1). This facility is maintained in the following models.

Gain modification and convergence problem



A small gain leads to smooth edges and output now seems to be non-symmetric. In order to get sharp edges as displayed above, gain must be increased. As our model uses a voltage source with a very high gain (1meg), we must be aware of difficulties encountered by the .OP analysis. So as to succeed, parameters Vmin and Vmax of the .OP analysis will be adapted, and fixed to larger values.

Second step: what about a non symmetric threshold?

If we now want to model a trigger, as a non-symmetric comparator, we can use a similar macro-model, where the main modification is the introduction of parameters vmin and vmax, which determines vin threshold. This step points out the ease of modification when using a functional macro-model description.

Please refer to the previous stage decomposition, which is similar to the present one, except for the slightly modified input stage. Indeed the differential input is not taken into account, as we want to point out the disymmetric aspect.

Modifications on the intermediate stage concern coefficient of the controlled voltage supply E1 as you can notice in the netlist file below.

```

*-----file schmitt.nsx-----
*SCHMITT TRIGGER from
* "MacroModeling with Spice".
*This comparator includes an
* hysteresis specified by VMIN and
* VMAX, respectively the minimum and
* the maximum voltage for the vout
* shifting.
*Vout is comprised between VOUTMIN
* and VOUTMAX.
*Parameter INV allows inversion of
* the output signal.
* INV = 1 : inverter trigger.
*      -1: non-inverter trigger.
*Note : No offset or differential
* input.

.SUBCKT SCH vin vout \
+ VMIN VMAX
+ VOUTMIN VOUTMAX RIN ROUT INV

R1  vin 0 RIN

E1 3 0
+ VALUE { 1e5 *((VMIN+VMAX)/2
+ - V(vin)
+ +(VMAX-VMIN)*V(4)/(VOUTMAX-VOUTMIN))}
R2 3 4 1MEG
C1 4 0 1e-15
D1 4 5 SWITCH
V1 5 0 '(VOUTMAX-VOUTMIN)/2'
D2 6 4 SWITCH
V2 6 0 '-(VOUTMAX-VOUTMIN)/2'
.model SWITCH D(IS=10u N=0.001)

EOUT 7 0 POLY(1) 4 0
+ 'VOUTMAX/2+VOUTMIN/2'
+ INV
R3 vout 0 1G
R4 7 vout ROUT
.ENDS SCH

*ideal case (no offset or hysteresis)
*Xideal vin voutideal \
++ 0.0 0.0 -5.0 5.0 1MEG 1K -1
++ SCHMITT

* Standard case with
* vmin=vmax=threshold
* to design an artificial voltage
* input offset.
Xstandard vin vout \
+ 0.5 0.5 -5.0 5.0 1MEG 1K -1
+ SCHMITT

* Hysteresis case with
* vshiftmin=-1
* vshiftmax=0.5
Xhysteresis vin vouthyster \
+ 0.0 0.5 -5.0 5.0 1MEG 1K -1
+ SCHMITT

*inverter (only INV is modified)
*Xinv vin voutinv \
++ 0.0 0.5 -5.0 5.0 1MEG 1K 1
++ SCHMITT

```

```

*-----file schmitt.pat----
*SOURCES
V_VIN VIN 0 SIN( 0 2 1MEG 0 0 0 )

*SCREEN
.TRACE TRAN V(VIN) V(VOUT)
V(VOUTHYSTER) MIN=-6.0003070E+000
MAX=6.0002938E+000

.OP EPS_V=1U VMIN=-5MEG VMAX=50MEG
DELTAV=0 EPS_I=100P MAXITER=5000
.EPS 1U 100M 1N
.H 10N 1F 10N 250M 2
.TRAN 10N 2U 0

```


Circuit description:

- Input stage
- Central stage
- Output stage

Simulation

By default standard and hysteresis cases are launched (other calls are preceded by a comment character).

In order to visualize ideal case and/or inverted case, you have to remove comment characters in front of the command lines and to run a transient analysis.

Disymmetric case.

The real interest of this model is the hysteresis function.

There is no direct access to the gain to force sharp edges. We can see that the output signal sharpness depends on parameters v_{min} and v_{max} which indicates the extreme v_{in} values for the shifting edges. Indeed the gain is set by the ratio $(v_{outmax}-v_{outmin})/(v_{max}-v_{min})$ but with a calculus on v_{in} instead of v_{out} , hence avoiding trouble when voltage threshold are equals.

You can easily observe this gain thanks to SMASHTM. Observe the output $v_{outhyster}$ using V_{in} as X-axis.

Third step: use of a behavioral model

The previous generic description is helpful but value of the voltage source E1 is quite strange at first glance, and in any case does not immediately appear to model an hysteresis. Furthermore, as hysteresis are common in electronic devices, why don't we use a generic model of hysteresis, which could be reused in other designs? This section will introduce you to analog behavioral modeling.

We model hysteresis thanks to a behavioral model, illustrating the four possible states of this device:

- $v_{out} = v_{outmax}$ if $v_{in} > v_{max}$: state 3.
- $v_{out} = v_{outmin}$ if $v_{in} < v_{min}$: state 1.
- if v_{in} between v_{min} and v_{max} , v_{out} depends on the previous output of the device (this is the reason for the parameter v_{loop} to toggle between states 2 and 4) as shown on the figure 4.




Figure 4: Detailed functional schematic

Based on the stage decomposition of figure 3, we now introduce a behavioral block in place of voltage source E2. We first describe the behavioral module (source code) and then list the netlist and pattern files.

Behavioral module.

DECLARATIONS:

```
INPUTS:      VIN VLOOP
OUTPUTS:     VOUT
PARAMS:      VMIN VMAX VOUTMIN VOUTMAX
```

BEHAVIOR:

```
{
/* HYSTERESIS
VIN is the actual differential input.
Vloop is the present output of the module, used to
find out the present state (2 or 4) when VIN is
between VMIN and VMAX.
Output is equal to VOUTMIN or VOUTMAX depending on
the following:
*/
/* State 1 */
if (VIN <= VMIN)
    VOUT = VOUTMIN;
/* State 2 */
if ((VIN > VMIN) && (VIN < VMAX) && (VLOOP < 0))
    VOUT = VOUTMIN;
/* State 3 */
if (VIN >= VMAX)
    VOUT = VOUTMAX;
/* State 4 */
if ((VIN > VMIN) && (VIN < VMAX) && (VLOOP > 0))
    VOUT = VOUTMAX;
}
```

Testing the behavioral module. Simulation files.

```

*-----file hystcomp.nsx---

.SUBCKT HYSTCOMP vin vref vout \
+ VIO VDOWN VUP GAIN
+ VOUTMIN VOUTMAX RIN ROUT INV

R1 vin vref RIN

E1 1 0 POLY(2) vin 0 vref 0
+ '-VIO*GAIN'
+ GAIN
+ '-GAIN'
Rtest1 1 0 1

* Call to behavioral module.
ZHYSSTERESIS
+ IN( 1 4 )
+ OUT( 4/ZS=1T,1e-15 )
+ PAR( 'VDOWN*GAIN' 'VUP*GAIN' -1e7 1e7 )
+ ZAHYSTER

D1 4 5 SWITCH
V1 5 0 '(VOUTMAX-VOUTMIN)/2'
D2 6 4 SWITCH
V2 6 0 '-(VOUTMAX-VOUTMIN)/2'
.model SWITCH D(IS=10u N=0.001)

EOUT 9 0 POLY(1) 4 0
+ 'VOUTMAX/2+VOUTMIN/2'
+ INV
R3 vout 0 1G
R4 9 vout ROUT
.ENDS

*standard case
Xtest vin vref vout \
+ 0.5 0.0 0.0 10.0 -5.0 5.0 1MEG 1K 1
+ HYSTCOMP

* Hysteresis case
*Xhyst vin vref vouthyster \
*+ 0.0 0.0 0.5 20.0 -5.0 5.0 1MEG 1K 1
*+ HYSTCOMP

```

```

*-----file hystcomp.pat---

*SOURCES
V_VIN VIN 0 SIN( 0 2 1MEG 0 0 0 )
V_VREF VREF 0 0

*SCREEN
.TRACE TRAN V(VIN) V(VOUT)
V(VOUTHYSTER) MIN=-5.9999955E+000
MAX=6.0000107E+000


* MODELE HYSTERESIS
.LIB ZAHYSTER.AMD

.OP EPS_V=1U VMIN=-30MEG VMAX=30MEG
DELTAV=2 EPS_I=100P MAXITER=300
.EPS 1U 100M 1N
.H 1N 100P 5N 250M 2
.TRAN 10N 2U 0

```

Results are acceptable but simulation is very slow.

Standard case and non-symmetric threshold.



This simple description implies sharp discontinuities of the vout signal. These discontinuities can explain the lack of rapidity of the behavioral model.

Last step: a continuous model for hysteresis function.

In this last part, only the behavioral model is modified, so as to get a continuous hysteresis function.

In order to speed up the simulation, we have to use a continuous output signal and we suggest to use the hyperbolic tangent function.

Doing so, we observed an oscillation of the output signal around the transition edge. Actually it was a consequence of the oscillation between both the “up” way and the “down” way, which is a direct transition between state 2 and state 4 of figure 4. In order to avoid such an oscillation, we must “remember” which curve the signal follows, and allow a change of curve only during states 1 and 3. The “up” way is called the curve 0 (rising edge from voutmin to voutmax), when the “down” way is called the curve 1. Allowed changes are:

- curve 0 to curve 1 when in state 3,
- curve 1 to curve 0 when in state 1.

To implement the memory of the module, we use a boolean variable, CURVE. As it is declared as a GLOBAL_BOOLEAN, its value is maintained between successive calls of the module.

Before listing the source code, we introduce hyperbolic tangent. This continuous function makes sharp transitions less sharp (!). In order to fit the ideal curve (dotted line on the figure 5), we must calculate the inflection point according to the different threshold values (voutmin, voutmax, vdown or vup depending on the considered curve).




Figure 5: Interest of hyperbolic tangent

Code of the behavioral module:

```

DECLARATIONS:

INPUTS:          VIN
OUTPUTS:         VOUT STATE
PARAMS:          VDOWN VUP VOUTMIN VOUTMAX GAIN
GLOBAL_BOOLEAN:  CURVE

BEHAVIOR:
{
/* HYSTERESIS with hyperbolic tangent
VIN is the actual differential input.
Choice between UP curve and DOWN curve (respectively curves 1 and 0)
is only possible in stable states (namely states 1 and 3). This
will avoid oscillations.
VDOWN and VUP are the input threshold voltages.
*/
double x;
double VINFLEXION, VINFLEX, VDINFLEX, VUPINFLEX;
double VINFL, VSUP;

/* Inflexion point
(VDinflex;Vinflexion) and (VUPinflex;Vinflexion).
Vinflex= ArgTan(Vinflexion/(VOUTMAX-VOUTMIN))*/
VINFLEXION = 0.5*(VOUTMAX+VOUTMIN);
VINFLEX = 0.5*log((3*VOUTMAX-VOUTMIN)/(VOUTMAX-3*VOUTMIN));
VDINFLEX = VINFLEX+VDOWN;
VUPINFLEX = VINFLEX+VUP;

/* Limits of unstable area (hysteresis)*/
if (VDINFLEX==VUPINFLEX){
    VINFL = VDINFLEX-0.5;
    VSUP = VDINFLEX+0.5;
}
else {
    VINFL = min(VDINFLEX,VUPINFLEX)-fabs((VDINFLEX-VUPINFLEX)/1.5);
    VSUP = max(VDINFLEX,VUPINFLEX)+fabs((VDINFLEX-VUPINFLEX)/1.5);
}

/* Initialisation: default curve=0 */
if ( FIRSTCALL ){
    CURVE = 0;
    x = (VIN-VUPINFLEX)/GAIN;
    TVINFL = VINFL;
    TVSUP = VSUP;
}
}

```

```

/* State 1 */
if (VIN <= VINFL) {
    CURVE = 0;
    x = (VIN-VUPINFLEX)/GAIN;
    VOUT = VOUTMIN;
    STATE = 1;
    return;
}

/* State 3 */
if (VIN >= VSUP) {
    CURVE = TRUE;
    x = (VIN-VDINFLEX)/GAIN;
    VOUT = VOUTMAX;
    STATE = 3;
    return;
}

/* State 4*/
if ((VIN >VINFL) && (VIN < VSUP) && (CURVE))
{
    CURVE = 1;
    x = (VIN-VDINFLEX)/GAIN;
    VOUT = ((VOUTMAX-VOUTMIN)/2)*tanh(x)+VINFLXION;
    SETDERIV( VOUT_Node, VIN_Node, (VOUTMAX-VOUTMIN)/2 );
    STATE = 4;
    return;
}


/* State 2*/
if ((VIN >VINFL) && (VIN < VSUP) && (!CURVE))
{
    x = (VIN-VUPINFLEX)/GAIN;
    VOUT = ((VOUTMAX-VOUTMIN)/2)*tanh(x)+VINFLXION;
    SETDERIV( VOUT_Node, VIN_Node, (VOUTMAX-VOUTMIN)/2 );
    CURVE = 0;
    STATE = 2;
    return;
}

/* Indetermined state */
STATE = 5;
VOUT = VOUTMAX;
}

```

Note that a return to state 1 by following the curve 0 is no longer forbidden and thus the hysteresis function does not behave exactly as expected. Nevertheless, the ideal case is well modeled and we can observe a true hysteresis. If the extreme states 1 and 3 are never reached (i.e. $v_{shiftmin} < v_{in} < v_{shiftmax}$), the output signal always follows an arbitrary default curve (namely curve 0).

Disymmetric case and visualization of STATE



This simulation is much faster than the previous one. It is thus interesting to write a behavioral model a little bit more complicated, but continuous, so as to gain speed...

Reference

- Ref. 1 “Macromodeling with spice” of J.Alvin Connelly and Pyung Choi
1992 by Prentice-Hall.